
ABSTRACT 
Publication bias is one threat to validity that researchers conducting meta-analysis studies confront.  Although 
statistical methods for detecting publication bias have surfaced in the literature (e.g., Begg Rank Correlation, Egger 
Regression, Funnel Plot Regression, and Trim & Fill), many researchers rely on visual inspection of funnel plots.  This 
program was created to provide meta-analysts with the ability to implement statistical methods to detect publication 
bias.  Statistical information from the corpus of studies (i.e., effect sizes and sample sizes) is input in this SAS macro 
and analyzed for potential publication bias using the above mentioned statistical methods.  In addition to the p-value 
associated with each detection method the program output provides the estimated mean effect size and random 
effects variance component.  The application of the SAS/IML programming on two sets of example data (with and 
without publication bias) is provided along with the macro programming language.   

INTRODUCTION 
Publication bias is an important issue that researchers face when conducting a literature review, designing a new 
study, or conducting a meta-analysis. Unfortunately, when a researcher gathers literature their findings do not include 
all studies that have occurred regarding the specified content area searched. This phenomenon was discussed by 
Rosenthal (1979) as the “file drawer problem” or publication bias. Essentially, researchers may have studies that are 
sitting in their filing cabinets because they decided not to publish or were rejected by journals. Reasons researchers 
do not submit studies or for journals to reject studies typically revolve around whether the results indicated significant 
findings, which are influenced by sample size, or large effects. In addition, published research can inadvertently 
contribute to publication bias when researchers exclude non-significant findings from results or report data poorly. 
Thus, there is a pattern in the published literature of a disproportionate number of studies with statistically significant 
findings and large effects. 
 
When meta-analysts do not include unpublished studies, the results of the meta-analysis may be biased. Specifically, 
the meta-analysis results may indicate an inflated effect because the published studies are more likely to have 
significant results and large effects (Sharpe, 1997). Thus, publication bias is considered to be a threat to the validity of 
meta-analyses. One method for detecting publication bias is the visual interpretation of a funnel plot (a scatterplot of 
effect sizes and sample sizes). However, visual examination of the funnel plot is limited because the interpretation is 
subjective and the plot can be difficult to interpret when there are a small number of studies included in the meta-
analysis (Greenhouse & Iyengar, 1994; Thornton & Lee, 2000). Consequently, some researchers have developed 
statistical methods for detecting publication bias that are not subjective. 
 
IMPACT OF PUBLICATION BIAS ON META-ANALYTIC SUMMARIES 
Using a Monte Carlo design Rendina-Gobioff (2006) examined the impact of moderate and strong publication bias on 
the estimated mean effect size and the estimated effect size variance results of random-effects meta-analyses.  
Consistent with the literature Rendina-Gobioff (2006) found that  when no publication bias was imposed the average 
effect size bias was -0.0120 with a minimum value of -0.1125 and maximum value of 0.0997. In contrast, the average 
effect size bias increased to 0.0792 when moderate publication bias was imposed, with a minimum value of -0.0327 
and maximum value of 0.3030. The average effect size bias increased even more when the imposed publication bias 
was strong, 0.1350 (minimum= -0.0295 and maximum=0.4491). According to these results when a researcher is 
conducting a meta-analysis with strong publication bias they could be producing an average effect size with as much 
as 0.45 error. 
 
Similar to bias associated with the mean effect size estimates, when there is no publication bias one would expect to 
have minimal effect size variance bias. Consistent with this assumption Rendina-Gobioff (2006) found that when no 
publication bias was imposed the average effect size variance bias was -0.0343 with a minimum value of -0.3806 and 
maximum value of 0.2756. In contrast, the average effect size variance bias increased to 0.1101 when moderate 
publication bias was imposed, with a minimum value of -0.1593 and maximum value of 0.7757. The average effect 
size variance bias increased even more when the imposed publication bias was strong, 0.2052 (minimum=-0.1413 
and maximum=1.1622). According to these results when researchers are conducting a meta-analysis with strong 
publication bias they could be producing an average effect size variance estimate with as much as 1.16 error. 

The findings presented by Rendina-Gobioff (2006) indicate that when meta-analyses are conducted with publication 
bias present the estimated mean effect size and the estimated effect size variance may include substantial error.  
Thus, there is a need for meta-analysts to have access to and to implement tools for detecting the presence of 
publication bias in their studies. 
 
STATISTICAL METHODS TO DETECT PUBLICATION BIAS 
The statistical methods for detecting publication bias included in this macro all examine the relationship between the 
effect sizes and the precision of the effect sizes. The various methods use different approaches for standardizing the 
effect size (or not standardizing it) and different definitions of precision (sample size, variance, inverse variance). 
However, the methods are all examining the relationship that is displayed in a funnel plot and the assumption that the 
absence of studies with small effect sizes and minimal precision (small sample size or large variance) provide 
evidence of publication bias. Therefore, a strong relationship is an indication of publication bias. An overview of the 
methods for detecting publication bias is presented in Table 1 along with the variables and analyses that are utilized 
with the method.  
 
Table 1 
Method for Detecting 
Publication Bias 

Variables from                    
Primary Studies Examined Analysis 

Funnel Plot  Effect Size Sample Size Visual Interpretation  
(non-statistical) 

Begg Rank Correlation (V) Standardized 
Effect Size 

Variance of       
Effect Size Rank Correlation 

Begg Rank Correlation (N) Standardized 
Effect Size Sample Size Rank Correlation 

Egger Regression  Standardized 
Effect Size Precision OLS Regression 

Funnel Plot Regression Effect Size Sample Size WLS Regression 

Trim and Fill 
Deviation of Effect 
Size from Mean 

Effect Size 

Number of studies 
included in Meta-

Analysis 

Nonparametric       
Rank Method 

Note. The standardized effect size included in the Begg Rank Correlation and Egger 
Regression analyses are calculated differently. 
 
 
Begg Rank Correlation method (Begg & Mazumdar, 1994) examines the relationship between the standardized 
treatment effect and the variance of the treatment effect using Kendall's Tau. The standardized treatment effects are 
estimated as: 

Ranks are assigned for the observed standardized treatment effects and the variances of those treatment effects 
(alternatively, the sample sizes may be ranked rather than the estimated variances). The correlation between these 
ranked values (Kendall’s Tau) leads to a statistical test for the presence of publication bias. 
 
The Egger Regression method (Egger, Smith, Schneider & Minder, 1997) treats the standardized treatment effect as 
the criterion and the precision of effect size estimation (the inverse of its standard error) as the predictor in a 
regression model (estimated by either OLS or WLS, with observations weighted by the inverse of their variances). 
The standardized treatment effects are estimated as: 

 Where gi is the observed study effect size for study i and vi is its variance. 

For the Egger Regression method, the precision of the effect size is estimated as: 
1

The Funnel Plot Regression method, suggested by Macaskill, Walter, and Irwig (2001), uses a regression model with 
the criterion variable being the treatment effect and study size being the predictor variable. Estimation by WLS is 

recommended for such a model, using as weights the inverse of the estimation variance (e.g., 1
iν
− ). In contrast to the 

Egger method, in this regression equation the slope will indicate no publication bias when it has the value of zero and 
the intercept in this regression equation will indicate the true effect. Thus, a test of the null hypothesis that the 
regression slope equals zero provides a test of publication bias. 
 
The Trim and Fill method, introduced by Duval and Tweedie (2000a, 2000b), is a nonparametric approach which is 
based on the funnel plot. Using symmetry assumptions the observed studies are ranked based on the absolute 
values of their deviations from the mean effect size; positive ranks for studies with effect sizes greater than the mean 
effect size, negative ranks for studies with effect sizes less than the mean effect size. The ranks are estimated as: 

( )*

A negative algebraic sign is assigned to ranks where the ig  is less than the 
ig  . Using these ranks the number of 

research studies missing from the funnel plot due to publication bias is estimated by: 
*

0 1R γ= −  
where R0 is the estimated number of studies concealed due to publication bias, 

* *
hk rγ = − , 

where k is the number of studies included in the meta-analysis, and *
hr  is the largest negative rank 

Publication bias is evidenced when R0 > 3, with power greater than 0.80 and α = .05 (Duval & Tweedie, 2000a). 
 
 
MACRO PUBBIAS 
A SAS/IML macro was designed to compute the Egger Regression, Begg Rank Correlation, Funnel Plot regression, 
and trim-and-fill tests for publication bias using either a fixed-effects or random-effects model for the meta-analysis. In 
addition, the macro provided by Mitchell (2000) that produces a forest plot of the effect sizes is incorporated, and a 
funnel plot is generated using SAS/GRAPH. The macro was developed to provide researchers with an easily 
accessible tool for conducting these tests and producing the plots. Arguments supplied to the macro include the name 
of the SAS dataset that contains the sample of effect sizes; the names of the variables for (a) sample sizes in each of 
the two groups, (b) the sample effect size, and (c) the study identification variable; and an indicator variable to use a 
fixed-effects (ModelType = 0) or random-effects (ModelType = 1) model for the computation of the publication bias 
tests.  
 
The output from the macro includes a table to present the results of the publication bias tests (Table 2), a forest plot 
of the effect sizes (Figure 1), and a funnel plot (Figure 2). 
 
PUBBIAS CODE 
%macro symsize; 
data _null_; 
set metadat; 
 retain sizeh1-sizeh%eval(&n_stud+2) 
 fontv1-fontv%eval(&n_stud+2); 
 length fontv1-fontv%eval(&n_stud+2) $ 20; 
 array sizes sizeh1-sizeh%eval(&n_stud+2); 
 array fvs $ fontv1-fontv%eval(&n_stud+2); 
 do i=1 to (&n_stud)+2; 
  if i=y then do; 
   sizes{i}=size*11/&maxsize; 
   fvs{i}='font=specialu v=K'; 
   if i=&n_stud+2 then output; 
  end; 
 end; 

 %do i=1 %to (&n_stud)+2; 
  call symput("sh&i",trim(left(put(sizeh&i,6.2)))); 
  call symput("fv&i",trim(left(put(fontv&i,20.)))); 
  %global sh&i fv&i; 
 %end; 
run; 
%mend symsize; 
%macro doall (study); 
 %do %while (&study le (&n_stud+2)); 
 if y=&study then do; 
  xx&study=lower95; 
  yy&study=y+0.2; output; 
  yy&study=y-0.2; output;  
  yy&study=y;  
  output;  
  xx&study=upper95; output;  
  yy&study=y+0.2;output; 
  yy&study=y-0.2; output; 
 end; 
 %let study=%eval(&study+1); 
 %end; 
%mend doall; 
 
%macro syms; 
 %do i=1 %to (&n_stud)+2; 
  symbol&i &&fv&i l=1 interpol=none h=&&sh&i color=black; 
 %end; 
%mend syms; 
%macro plotpts; 
 %do i=1 %to (&n_stud + 2); 
  yy&i*xx&i=%eval(&n_stud+3) 
 %end; 
 %do i=1 %to (&n_stud + 2); 
  y&i*x&i=&i 
 %end; 
%mend plotpts; 
%Macro PubBias(N1=size1,N2=size2,di=effsize,studyID=study,ModelType=1,Dataset=MetaPub); 
 
proc iml; 
 
 * +----------------------------------------------------------------------------------------+ 
     Read data from regular SAS into IML 
   +----------------------------------------------------------------------------------------+; 
use &Dataset; 
 read all var{&N1} into n1_vec; 
 read all var{&N2} into n2_vec; 
 read all var{&di} into di_vec; 
 read all var{&n1 &n2} into n_vec; 
 k= nrow(di_vec); 
 
* +----------------------------------------------------------------------------------------+ 
   Subroutine to calculate weighted mean effect size,  
   standard error, and confidence interval for mean. 
   Inputs to the subroutine are 
     di_vec - column vector of effect sizes (d) 
     var_di - column vector of estimation errors 
     tau2 - scalar estimate of RANDOM EFFECTS variance (set to 0 if Modeltype-fixed(0)) 
 
   Outputs are 
     d_mean = weighted mean d value  
     resum_wt = scalar, sum of the weights 

     vi_star = column vector of total variance for each study 
     d_SE = standard error of d 
     upper95, lower95 = endpoints of 95% CI 
 +----------------------------------------------------------------------------------------+; 
start mean_d(di_vec,var_di,tau2,d_mean,resum_wt,vi_star,d_SE,upper95,lower95); 
 
  k = nrow(di_vec); 
  d_mean = 0; 
  resum_wt = 0; 
  vi_star = J(k,1,0); 
  do i = 1 to k; 
    d_mean = d_mean + di_vec[i,1]/(var_di[i,1]+tau2);  
    resum_wt = resum_wt + (var_di[i,1]+tau2)##-1;  
 vi_star[i,1] = var_di[i,1]+tau2;   
  end; 
  d_mean = d_mean/resum_wt;  
  d_SE = SQRT(resum_wt##-1); 
  upper95 = d_mean + 1.96#d_SE; 
  lower95 = d_mean - 1.96#d_SE; 
finish; 
 
* +----------------------------------------------------------------------------------------+ 
  Subroutine to calculate the Q test 
   of homogeneity. 
   Inputs to the subroutine are 
     di_vec - column vector of effect sizes (d) 
     n_vec  - matrix (k X 2) of sample sizes 
              corresponding to each effect size 
 
   Outputs are 
     QQ = the obtained value of Q 
     prob_qq1 = chi-square probability associated with QQ 
     var_di = column vector of variances of effect sizes  
  +----------------------------------------------------------------------------------------+; 
start calcq(di_vec,n_vec,qq,prob_qq1,var_di); 
 
  k = nrow(di_vec); 
  var_di=J(k,1,0); 
  do i = 1 to k; 
    var_di[i,1] = ((n_vec[i,1]+n_vec[i,2])/(n_vec[i,1]#n_vec[i,2])) + 
((di_vec[i,1]##2)/(2#(n_vec[i,1]+n_vec[i,2]))); 
  end; 
 
  d_plus = 0; 
  fesum_wt = 0; 
  do i = 1 to k; 
    d_plus = d_plus + di_vec[i,1]/var_di[i,1];   
    fesum_wt = fesum_wt + var_di[i,1]##-1;       
  end; 
  d_plus = d_plus/fesum_wt;   
 
  QQ = 0; 
  do i = 1 to k; 
    QQ = QQ + ((di_vec[i,1] - d_plus)##2/var_di[i,1]); 
  end; 
 
  prob_qq1 = 1 - PROBCHI(QQ,k-1); 
finish; 
 
*+----------------------------------------------------------------------------------------+   
Subroutine to calculate OLS and WLS tests of models. 



6 

   Inputs to the subroutine are 
     di_vec - column vector of effect sizes (d) 
     n_vec  - matrix (k X 2) of sample sizes 
              corresponding to each effect size 
     X_Matrix - Matrix of potential moderator variables 
                For tests of publication bias, vi are predictors 
  vi - reciprocals of variances 
 
   Outputs are 
     B_wls - regression weights for WLS 
     SE_B  - Standard errors of the WLS weights 
     B_ols - regression weights for OLS 
     SE_B_ols  - Standard errors of the OLS weights 
  +----------------------------------------------------------------------------------------+; 
start calcreg(di_vec,n_vec,X_Matrix,vi,B_wls,SE_B,B_ols,SE_B_ols); 
 
  k = nrow(di_vec); 
  X = J(k,1,1)||X_Matrix; 
  B_wls = INV(X`*DIAG(vi)*X)*X`*DIAG(vi)*di_vec; 
  cov_b = INV(X`*DIAG(vi)*X); 
  SE_B = SQRT(vecdiag(cov_b)); 
 
  B_ols =INV(X`*X)*X`*di_vec; 
  cov_b = INV(X`*X); 
  SE_B_ols = SQRT(vecdiag(cov_b)); 
finish; 
 
* +----------------------------------------------------------------------------------------+ 
   Subroutine Kendall 
    Computes the Kendall Tau for Norman Cliff ordinal level analyses. 
    Arguments to the subroutines are: 
     A B = vectors of observed data for the two variables  
  (A will be the observed tx effects 
    B will be the variance of the tx effects) 
     N = sample size 
 
   Returned are: 
    T_AB = Kendall Tau Coefficient Y and X1 
    UNTIE_A = proportion of scores that are not tied on A 
    UNTIE_B = proportion of scores that are not tied on B 
    VART_AB = VARIANCE of Y AND X1 
    Z_TEST = obtained value of Z for test of Kendall Tau Coefficient 
  +----------------------------------------------------------------------------------------+; 
START KENDALL(A,B,N,T_AB,untie_A,untie_B,VART_AB,Z_TEST); 
 
 DOM_MTXA = J(N,N,0); 
 
 ties_A   = 0; 
 counts_A = 0; 
 do i = 1 to N; 
    do j = 1 to N; 
      if A[i,1] > A[j,1] then do; 
           DOM_MTXA[i,j] =  1; 
      end; 
      if A[i,1] < A[j,1] then do; 
           DOM_MTXA[i,j] = -1; 
      end; 
      if A[i,1] = A[j,1] then do; 
           ties_A = ties_A + 1; 
      end; 
      counts_A = counts_A + 1; 

    end; 
 end; 
 untie_A = 1 - (ties_A - N)/(counts_A - N); 
 
 DOM_MTXB = J(N,N,0); 
 
 ties_B   = 0; 
 counts_B = 0; 
 do i = 1 to N; 
    do j = 1 to N; 
      if B[i,1] > B[j,1] then do; 
           DOM_MTXB[i,j] =  1; 
      end; 
      if B[i,1] < B[j,1] then do; 
           DOM_MTXB[i,j] = -1; 
      end; 
      if B[i,1] = B[j,1] then do; 
           ties_B = ties_B + 1; 
      end; 
      counts_B = counts_B + 1; 
    end; 
 end; 
 untie_B = 1 - (ties_B - N)/(counts_B - N); 
 
 DOM_MTXD = J(N,N,0); 
 
 do i = 1 to N; 
    do j = 1 to N; 
      DOM_MTXD[i,j] = DOM_MTXA[i,j]#DOM_MTXB[i,j]; 
    end; 
 end; 
 
 MTXD_sum = DOM_MTXD[,+]; 
 
 T_AB = MTXD_sum[+,] #(1/(n#(n-1))); 
 
MTX_F = J(N,1,0); 
do i = 1 to N; 
     MTX_F[i,1] = (MTXD_sum [i,1]/(n-1)); 
 end; 
 
MTX_G = J(N,1,0); 
 do i = 1 to N; 
     MTX_G[i,1] = (MTX_F[i,1] - T_AB)##2; 
 end; 
 
 MTXG_sum = 1/(n-1)#(MTX_G[+,]); 
 
MTX_H = J(N,N,0); 
 do i = 1 to N; 
    do j = 1 to N; 
       MTX_H[i,j] = DOM_MTXD[i,j]#DOM_MTXD[i,j]; 
    end; 
 end; 
 
 MTXH_sum = MTX_H[+,+]; 
 
 NUMER_AB = MTXH_sum - ((n) # (n-1) # (T_AB#T_AB)); 
 DENOM_AB = n#(n-1) -1; 
 
 VART_AB = NUMER_AB/DENOM_AB; 

 vart_ab = ((4#(n-2)#(MTXG_sum))+(2#VART_AB)) / (n#(n-1)); 
 
IF vart_ab > 0 THEN DO ; 
 Z_TEST = (T_AB /SQRT(vart_ab)); 
end; 
 
if vart_ab =0 then do; 
 Z_TEST = 5.00; 
END; 
 
FINISH; 
 
* +----------------------------------------------------------------------------------------+   
 Main program 
  +----------------------------------------------------------------------------------------+; 
 
* +----------------------------------------------------------------------------------------+
 Calculate Q 
  +----------------------------------------------------------------------------------------+; 
   run calcq(di_vec,n_vec,qq,prob_qq1,var_di); 
 
        CC = (J(1,K,1)*var_di##-1) - ((J(1,K,1)*var_di##-2) /  
                                    (J(1,K,1)*var_di##-1)); 
     Tau2 = (QQ - (K - 1)) / cc;  
   if tau2 < 0 then tau2 = 0; 
   if &modeltype=0 then tau2=0;  
* +----------------------------------------------------------------------------------------+ 
    Calculate weighted mean effect size and confidence interval 
  +----------------------------------------------------------------------------------------+; 
   run mean_d(di_vec,var_di,tau2,d_mean,resum_wt,vi_star,d_SE,upper95,lower95); 
 
  vi = J(k,1,0); 
  vi_inv = j(k,1,0); 
  mean = 0; 
  weight = 0; 
  t_star = J(k,1,0); 
  Vi_stand = J(k,1,0); 
  dev_di = J(k,1,0); 
  Root_Vi=J(k,1,0); 
  Egger_z = J(k,1,0); 
  REJ_TRIM = J(3,1,0); 
 
  do i = 1 to k; 
    vi[i,1] = vi_star[i,1];       
 vi_inv[i,1] = vi[i,1]##-1;      
 mean = d_mean;         
 weight = resum_wt;       
    dev_di[i,1] = ABS(di_vec[i,1] - mean);  
    Vi_stand[i,1] = vi[i,1] - (weight##-1); 
    t_star[i,1] = (di_vec[i,1] - mean)/SQRT(Vi_stand[i,1]);  
 Root_Vi[i,1] = vi[i,1]##-0.5;    
 Egger_z[i,1] = di_vec[i,1]#Root_Vi[i,1]; 
 
  end; 
* +----------------------------------------------------------------------------------------+ 
    Calculate Egger OLS Regression (Precision as predictor) 
  +----------------------------------------------------------------------------------------+; 
run calcreg(Egger_z,n_vec,Root_Vi,vi_inv,B_wls,SE_B,B_ols,SE_B_ols); 
 
  eggt_ols = B_ols[1,1]/SE_B_ols[1,1]; 

  eggPR_tOLS =2#(1-probt(abs(eggt_ols),k-2)); 
* +----------------------------------------------------------------------------------------+   
 Calculate Begg Rank Correlation(Variance as predictor) 
  +----------------------------------------------------------------------------------------+; 
  run KENDALL(t_star,vi,k,T_X1Y,UT_A,UT_B,VART_X1Y,Z_TEST); 
  BeggV_z = Z_TEST; 
  BeggVpr_z =2#(1-probnorm(abs(Z_test))); 
* +----------------------------------------------------------------------------------------+ 
    Calculate Begg Rank Correlation (Sample size as predictor) 
 +----------------------------------------------------------------------------------------+; 
  total_n = n_vec * J(2,1,1); 
  run KENDALL(t_star,total_n,k,T_X1Y,UT_A,UT_B,VART_X1Y,Z_TEST); 
  BeggN_z = Z_TEST; 
  BeggNpr_z =2#(1-probnorm(abs(Z_test))); 
* +----------------------------------------------------------------------------------------+ 
    Calculate Funnel Plot WLS Regression (Sample size as predictor) 
  +----------------------------------------------------------------------------------------+; 
  run calcreg(di_vec,n_vec,total_n,vi_inv,B_wls,SE_B,B_ols,SE_B_ols); 
 
  Funt_WLS = B_wls[2,1]/SE_B[2,1]; 
  FunPR_tWLS =2#(1-probt(abs(Funt_WLS),k-2)); 
* +----------------------------------------------------------------------------------------+ 
    Calculate Trim and Fill 
  +----------------------------------------------------------------------------------------+; 
  dev_rank = rank(dev_di);  
  do i = 1 to k; 
   if di_vec[i,1] < mean then dev_rank[i,1] = -1#dev_rank[i,1]; 
  end; 
  r=-1#MIN(dev_rank); 
  gamma=k-r; 
  ro=gamma-1; 
  r2=MAX(dev_rank); 
  gamma2=k-r2; 
  ro2=gamma2-1; 
  if ro > 3 then REJ_TRIM[1,1] = REJ_TRIM[1,1] +1; 
  if ro2 > 3 then REJ_TRIM[2,1] = REJ_TRIM[2,1] +1; 
   * +------------------------------------------------------------------+ 
    Either tail (note: alpha is .10 for this 
     +------------------------------------------------------------------+; 
  if (ro > 3 | ro2 > 3) then REJ_TRIM[3,1] = REJ_TRIM[3,1] +1; 
 
  Right=REJ_TRIM[1,1]; If Right=1 then Right='Yes'; Else Right='No'; 
  Left=REJ_TRIM[2,1];If Left=1 then Left='Yes'; Else Left='No'; 
  Both=REJ_TRIM[3,1];If Both=1 then Both='Yes'; Else Both='No'; 
  
* +----------------------------------------------------------------------------------------+ 
    Print output results 
  +----------------------------------------------------------------------------------------+; 
 
file print; 
put  
@1'-------------------------------------------------------------------------------------'// 
@1 'Meta-Analysis: Descriptive Information'/ 
@1 '--------------------------------------'// 
@5'Number of Studies'@60 k// 
@5'Mean Effect Size' @60 d_mean 10.4/ 
@7'Confidence Band'/ 
@10 '95% Lower Limit'  @60lower95 10.4/ 
@10'95% Upper Limit'@60 upper95 10.4// 
@5 'Test for Homogeneity'@60'Chi Square'@75'Probability'/ 
@5 '--------------------' @60'----------'@75'-----------'/ 

@5'Q test'@62 QQ 8.4 @75 prob_qq1 10.4// 
@5'Random Effects Variance Component'@62 Tau2 8.4/// 
@1'--------------------------------------------------------------------------------------'// 
@1 'Meta-Analysis: Tests of Publication Bias'/ 
@1 '----------------------------------------'// 
@5 'Egger Regression' @60't value'@75'Probability'/ 
@5 '----------------' @60'-------'@75'-----------'/ 
@5 'Egger OLS Regression' @60 eggt_OLS 7.4 @75 eggPR_tOLS 10.4/// 
@5 'Begg Rank Correlation' @60'z value'@75'Probability'/ 
@5 '---------------------' @60'-------'@75'-----------'/ 
@5 'Begg Rank Correlation (Predictor=Variance)' @60 beggV_z 7.4 @75 beggVpr_z 10.4/ 
@5 'Begg Rank Correlation (Predictor=Sample Size)' @60 beggN_z 7.4 @75 beggNpr_z 10.4/// 
@5 'Funnel Plot Regression' @60't value'@75'Probability'/ 
@5 '----------------------' @60'-------'@75'-----------'/ 
@5 'Funnel Plot WLS Regression' @60 Funt_WLS 7.4  @75 FunPR_tWLS 10.4/// 
@5 'Trim and Fill'@60'Publication Bias Present'/ 
@5 '-------------'@60'------------------------'/ 
@7 'Right Tail' @70 Right/ 
@7 'Left Tail' @70 Left/ 
@7 'Both Tails' @70 Both/ 
@1'--------------------------------------------------------------------------------------'//; 
 
* +----------------------------------------------------------------------------------------+ 
   Send summary information to regular SAS for plots 
  +----------------------------------------------------------------------------------------+; 
 sum_n = n_vec*J(2,1,1); 
 total_n = sum(sum_n); 
 outvector = d_mean||lower95||upper95||total_n; 
 create j1 from outvector;  
 append from outvector; 
quit; 
 
data meta2; 
 set &Dataset; 
 SE = SQRT((&n1 + &n2) / (&n1 * &n2) + &di**2 / (2*(&n1 + &n2))); 
 lower95 = &di - 1.96*SE; 
 upper95 = &di + 1.96*SE; 
 size = &n1 + &n2; 
 y = _n_; 
 
data total1; 
 set j1; 
 rename col1 = &di col2 = lower95 col3 = upper95 col4 = size;  
 length &StudyID $ 12; 
 y = 99; 
 &StudyID = 'Total'; 
data total2; 
 y = 98; 
data total; 
 set total1 total2; 
 axis1 label=(height=2.9 
 font='Zapf' 'Effect Size') 
 minor=none 
 value=(height=2.5 font='Zapf'); 
 axis2 label=(height=2.9 angle = 90  
 font='Zapf' 'Sample Size') 
 minor=none 
 value=(height=2.5 font='Zapf'); 
 title1 font = 'Zapf' height = 2.9 'Funnel Plot' ; 
 symbol1 interpol=none 
       value=circle 

       height=3 
       cv=black 
       ci=black 
       co=black 
       width=2; 
 
proc gplot data=meta2; 
 plot size*&di /haxis=axis1 vaxis=axis2 frame ; 
run; 
 
proc means data=meta2 noprint; 
 var size; 
 output out=meanout sum=sum max=max; 
data _null_; 
set meanout end=eof; 
if eof then do; 
 call symput("n_stud",trim(left(put(_freq_,8.)))); 
 call symput("n_subs",trim(left(put(sum,8.)))); 
 call symput("maxsize",trim(left(put(max,8.)))); 
end; 
 
proc sort data=meta2; by &di upper95; 
run; 
 
data metadat; 
set total(in=a) meta2(in=b); 
 length metastr $ 200; 
 retain metastr studcnt; 
 metanum=y; 
 y=_n_; 
 if _n_=1 then do; 
  metastr="0=' ' %eval(&n_stud+3)=' '"; 
  studcnt=&n_stud+2; 
 end; 
 else studcnt=studcnt-1; 
 do i=1 to %eval(&n_stud+2); 
  if i=studcnt then metastr=" " || trim(left(metastr)) || ' ' || trim(left(y)) 
||"='" || trim(left(&StudyID)) || "' "; 
 end; 
 if &StudyID=' ' then y=.; 
 yo=lower95; xo=upper95; 
 output metadat; 
 if _n_=%eval(&n_stud+2) then do; 
  call symput("metastr",trim(metastr)); 
 end; 
run; 
proc format; 
 value stfmt &metastr; 
run; 
 
 
data alldata; 
 set metadat; 
 %doall (1) 
 %symsize 
 %syms 
 
data final(drop=i); 
set alldata; 
 array xarray{*} x1-x%eval(&n_stud+2); 
 array yarray{*} y1-y%eval(&n_stud+2); 

 do i=1 to (&n_stud)+2; 
  if i=y then do; 
   xarray{i}=&di; 
   yarray{i}=y; 
  end; 
 end; 
format yy1-yy%eval(&n_stud+2) stfmt.; 
run; 
 axis1 label=(height=2.9 
 font='Zapf' 'Effect Size') 
 minor=none 
 value=(height=2.5 font='Zapf'); 
 axis2 label=(height=2.9 angle = 90  
 font='Zapf' 'Study') 
 minor=none 
 order=( 0 to %eval(&n_stud+3) by 1) 
 value=(height=2.5 font='Zapf'); 
 title1 font = 'Zapf' height = 2.9 'Forest Plot' ; 
 
proc gplot data=final; 
plot  
%plotpts / overlay haxis=axis1 vaxis=axis2 frame href=0; 
symbol%eval(&n_stud+3) f=marker v=none l=1 w=1 i=join; 
symbol1 font=marker v=P l=1 h=2.7 interpol=none; 
run; 
%Mend PubBias; 
 
 
EXAMPLE OF MACRO PUBBIAS 
The easiest way in which the macro PubBias may be used is to simply create a SAS dataset that inputs the sample 
effect sizes, sample sizes, and study identification information that are to be included in the meta-analytic model. The 
macro is then called, using as arguments the name of the dataset and the names of the relevant variables. Summary 
data from 10 studies are used to illustrate the macro. The sample sizes, effect sizes, and study identification 
information are read into the SAS data set MetaPub. 
 
Data MetaPub; 
 Input size1 1-3 size2 5-7 effsize 9-12 study $ 14 - 25; 
 Datalines; 
 25  30 0.75 Able 1986    
100 125 0.45 Bonk 1994    
250 250 0.25 Carson 1990  
 90 150 0.35 Diddle 1993  
 50  60 0.70 Efron 1991   
180 130 0.39 Flipper 1989 
 68  82 0.50 Goober 1975  
170 200 0.30 Halcyon 2002 
110  90 0.42 Illy 2004    
 45  45 0.65 Jersey 2002  
 ; 
 
The following call to the macro identifies the variables SIZE1 and SIZE2 as the sample sizes for each group, 
EFFSIZE as the effect size, STUDY as the study identification variable, and METAPUB as the SAS data set 
containing the information. Finally, the ModelType = 1 argument requests a random-effects model for the publication 
bias analysis. 
 
%PubBias(N1=size1,N2=size2,di=effsize,studyID=study,ModelType=1,Dataset=MetaPub); 
run; 
 
OUTPUT FROM MACRO PUBBIAS 
Table 2 provides an example of the tabled output produced by the macro PubBias. The SAS/Graph output from the 
macro is provided in Figures 1 and 2.  

The results presented in the upper half of Table 2 provide descriptive information about the effect sizes included in 
the meta-analysis.  In this case the meta-analysis consists of 10 studies with a random effects estimated mean effect 
size of 0.3922 with a 95% confidence band ranging from 0.3049 to 0.4795.  The effect sizes are considered 
homogeneous with a chi square value of 9.4967 (p>0.05).  A minimal amount of random effects variance was 
observed in these data (REVC=0.0011). 
 
The lower half of Table 2 presents the results of the statistical methods to detect publication bias.  The Egger 
Regression method positively detects publication bias (t = 2.8583, p<0.05).  Both of the Begg Rank Correlations 
methods positively detect publication bias.  Specifically, Begg Rank Correlation with variance as the predictor has a z 
value of 6.8750 (p<0.05) and the Begg Rank Correlation with sample size as the predictor has a z value of -6.8750 
(p<0.05).  The Funnel Plot Regression method also positively detects publication bias (t = -2.6961, p<0.05).  The Trim 
and Fill method is the only method not to indicate publication bias with these data.  All three indicators (right tail, left 
tail, and both tails) indicate no publication bias in the data. 
 
Table 2 
------------------------------------------------------------------------------------- 
Meta-Analysis: Descriptive Information 
-------------------------------------- 
    Number of Studies                                             10 

    Mean Effect Size                                           0.3922 
      Confidence Band 
         95% Lower Limit                                       0.3049 
         95% Upper Limit                                       0.4795 

    Test for Homogeneity                                   Chi Square     Probability 
    --------------------                                   ----------     ----------- 
    Q test                                                     9.4967         0.3927 
    Random Effects Variance Component                          0.0011 
-------------------------------------------------------------------------------------- 
Meta-Analysis: Tests of Publication Bias 
---------------------------------------- 
    Egger Regression                                       t value        Probability 
    ----------------                                       -------        ----------- 
    Egger OLS Regression                                    2.8583            0.0212 
 
    Begg Rank Correlation                                  z value        Probability 
    ---------------------                                  -------        ----------- 
    Begg Rank Correlation (Predictor=Variance)              6.8750            0.0000 
    Begg Rank Correlation (Predictor=Sample Size)          -6.8750            0.0000 
 
    Funnel Plot Regression                                 t value        Probability 
    ----------------------                                 -------        ----------- 
    Funnel Plot WLS Regression                             -2.6961            0.0272 
 
    Trim and Fill                                          Publication Bias Present 
    -------------                                          ------------------------ 
      Right Tail                                                     No 
      Left Tail                                                      No 
      Both Tails                                                     No 
-------------------------------------------------------------------------------------- 
 
Figure 1 presents a forest plot of the effect sizes and their confidence intervals for each study and the total (from the 
macro components provided by Mitchell, 2000).  The size of the point estimate box indicates the sample size of the 
study.  The studies with the larger sample sizes have smaller bands and larger boxes around the effect size.  In 
addition, the studies with larger sample sizes are closer to the total mean effect size estimate band.   
 
Figure 2 presents a funnel plot of the effect sizes included in the meta-analysis.  The funnel plot has a gap where 
studies with small samples sizes and small effect sizes are not present.  This is one indicator of publication bias.  

Figure 1. Forest Plot of Effect Sizes 
 

 
Figure 2. Funnel Plot of Effect Sizes 

In the following dataset (MetaPub2), the relationship between the effect sizes and the sample sizes is much weaker. 
The same call to the macro (except specifying Dataset = MetaPub2) yields the results presented in Table 3 and the 
graphs in Figures 3 and 4. For these data, none of the indices suggest publication bias. 
 
Data MetaPub2; 
 Input size1 1-3 size2 5-7 effsize 9-12 study $ 14 - 25; 
 Datalines; 
 25  30 0.75 Able 1986    
100 125 0.30 Bonk 1994    
250 250 0.50 Carson 1990  
 90 150 0.35 Diddle 1993  
 50  60 0.70 Efron 1991   
180 130 0.39 Flipper 1989 
 68  82 0.25 Goober 1975  
170 200 0.45 Halcyon 2002 
110  90 0.42 Illy 2004    
 45  45 0.65 Jersey 2002  
 ; 
 
%PubBias(N1=size1,N2=size2,di=effsize,studyID=study,ModelType=1,Dataset=MetaPub2); 
run; 
 
Table 3 
------------------------------------------------------------------------------------- 
Meta-Analysis: Descriptive Information 
-------------------------------------- 
    Number of Studies                                             10 

    Mean Effect Size                                           0.4380 
      Confidence Band 
         95% Lower Limit                                       0.3537 
         95% Upper Limit                                       0.5223 

    Test for Homogeneity                                   Chi Square     Probability 
    --------------------                                   ----------     ----------- 
    Q test                                                     7.4112         0.5944 
    Random Effects Variance Component                          0.0000 
-------------------------------------------------------------------------------------- 
Meta-Analysis: Tests of Publication Bias 
---------------------------------------- 
    Egger Regression                                       t value        Probability 
    ----------------                                       -------        ----------- 
    Egger OLS Regression                                    0.9201            0.3844 

    Begg Rank Correlation                                  z value        Probability 
    ---------------------                                  -------        ----------- 
    Begg Rank Correlation (Predictor=Variance)              0.3580            0.7203 
    Begg Rank Correlation (Predictor=Sample Size)          -0.3580            0.7203 

    Funnel Plot Regression                                 t value        Probability 
    ----------------------                                 -------        ----------- 
    Funnel Plot WLS Regression                              0.1044            0.9195 

    Trim and Fill                                          Publication Bias Present 
    -------------                                          ------------------------ 
      Right Tail                                                     No 
      Left Tail                                                      No 
      Both Tails                                                     No 
-------------------------------------------------------------------------------------- 

Figure 3. Forest Plot of Effect Sizes with No Publication Bias 
 

 
Figure 4. Funnel Plot of Effect Sizes with No Publication Bias 

CONCLUSION 
Meta-analysis has become increasingly important for the synthesis of research results in a variety of fields, including 
education, the behavioral sciences and medicine. However, the accuracy of inferences derived from meta-analysis 
may be threatened by the presence of publication bias. As the use of meta-analytic methods becomes more 
commonplace, researchers must remain mindful of the need to screen their samples of effect sizes for publication 
bias.   
 
The macro PubBias is provided to facilitate researchers’ calculation and use of common methods for testing for 
publication bias in meta-analysis. Although the macro, as provided, is limited to the analysis of standardized mean 
differences as effect sizes (i.e., Hedges g), the code is easily modified for the analysis of other indices of effect 
magnitude. For example, the analysis of Pearson Product Moment Correlation Coefficients requires a modification of 
the calculation of the variance in these effect sizes, and the incorporation of Fisher’s z transformation to normalize the 
sampling distribution of r. Further, simple modifications of the macro will produce 90% confidence intervals or one-
sided confidence intervals for the forest plot. 

